NAG Fortran Library Routine Document

D02JBF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

D02JBF solves a regular linear two-point boundary value problem for a system of ordinary differential equations by Chebyshev-series using collocation and least-squares.

2 Specification

```
 SUBROUTINE D02JBF(N, CF, BC, X0, X1, K1, KP, C, IC, W, LW, IW, LIW,
1                  IFAIL)
 INTEGER           N, K1, KP, IC, LW, IW(LIW), LIW, IFAIL
 real              CF, X0, X1, C(IC,N), W(LW)
 EXTERNAL          CF, BC
```

3 Description

This routine calculates the solution of a regular two-point boundary value problem for a regular linear nth-order system of first-order ordinary differential equations in Chebyshev-series in the range (x_0, x_1). The differential equation

$$y' = A(x)y + r(x)$$

is defined by the user-supplied function CF and the boundary conditions at the points x_0 and x_1 are defined by the user-supplied routine BC (see Section 5).

The user specifies the degree of the Chebyshev-series required, K1 − 1, and the number of collocation points, KP. The routine sets up a system of linear equations for the Chebyshev coefficients, n equations for each collocation point and one for each boundary condition. The boundary conditions are solved exactly, and the remaining equations are then solved by a least-squares method. The result produced is a set of coefficients for a Chebyshev-series solution for each component of the solution of the system of differential equations on a range normalised to $(-1, 1)$.

E02AKF can be used to evaluate the components of the solution at any point on the interval (x_0, x_1) – see Section 9 for an example. E02AHF followed by E02AKF can be used to evaluate their derivatives.

4 References

Picken S M (1970) Algorithms for the solution of differential equations in Chebyshev-series by the selected points method *Report Math. 94* National Physical Laboratory

5 Parameters

1: N – INTEGER *Input*

On entry: the order of the system of differential equations, n.

Constraint: N $\geq$ 1.

2: CF – ***real*** FUNCTION, supplied by the user. *External Procedure*

CF defines the system of differential equations (see Section 3). It must return the value of a coefficient function $a_{i,j}(x)$, of A, at a given point x, or of a right-hand side function $r_i(x)$ if J = 0.

Its specification is:

```
real FUNCTION CF(I, J, X)
INTEGER          I, J
real             X
```

1: I – INTEGER *Input*

2: J – INTEGER *Input*

On entry: indicate the function to be evaluated, namely $a_{i,j}(x)$ if $1 \leq \mathrm{J} \leq n$, or $r_i(x)$ if $\mathrm{J} = 0$.
$1 \leq \mathrm{I} \leq n$, $0 \leq \mathrm{J} \leq n$.

3: X – ***real*** *Input*

On entry: the point at which the function is to be evaluated.

CF must be declared as EXTERNAL in the (sub)program from which D02JBF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

3: BC – SUBROUTINE, supplied by the user. *External Procedure*

BC defines the n boundary conditions, which have the form $y_k(x_0) = s$ or $y_k(x_1) = s$. The boundary conditions may be specified in any order.

Its specification is:

```
SUBROUTINE BC(I, J, RHS)
INTEGER        I, J
real           RHS
```

1: I – INTEGER *Input*

On entry: the index of the boundary condition to be defined.

2: J – INTEGER *Output*

On exit: J must be set to $-k$ if the ith boundary condition is $y_k(x_0) = s$, or to $+k$ if it is $y_k(x_1) = s$.

J must not be set to the same value k for two different values of I.

3: RHS – ***real*** *Output*

On exit: the value s.

BC must be declared as EXTERNAL in the (sub)program from which D02JBF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

4: X0 – ***real*** *Input*

5: X1 – ***real*** *Input*

On entry: the left- and right-hand boundaries, x_0 and x_1, respectively.

Constraint: $\mathrm{X1} > \mathrm{X0}$.

6: K1 – INTEGER *Input*

On entry: the number of coefficients to be returned in the Chebyshev-series representation of the components of the solution (hence the degree of the polynomial approximation is K1 − 1).

Constraint: $\mathrm{K1} \geq 2$.

7: KP – INTEGER *Input*

On entry: the number of collocation points to be used.

Constraint: KP $\geq$ K1 $-$ 1.

8: C(IC,N) – ***real*** array *Output*

On exit: the computed Chebyshev coefficients of the kth component of the solution, y_k; that is, the computed solution is:

$$y_k = \sum_{i=1}^{\mathrm{K1}}{}' \mathrm{C}(i,k) T_{i-1}(x), \quad 1 \leq k \leq n$$

where $T_i(x)$ is the ith Chebyshev polynomial of the first kind, and $\sum'$ denotes that the first coefficient, $\mathrm{C}(1,k)$, is halved.

9: IC – INTEGER *Input*

On entry: the first dimension of the array C as declared in the (sub)program from which D02JBF is called.

Constraint: IC $\geq$ K1.

10: W(LW) – ***real*** array *Workspace*
11: LW – INTEGER *Input*

On entry: the dimension of the array W as declared in the (sub)program from which D02JBF is called.

Constraint: LW $\geq 2 \times \mathrm{N} \times (\mathrm{KP} + 1) \times (\mathrm{N} \times \mathrm{K1} + 1) + 7 \times \mathrm{N} \times \mathrm{K1}$.

12: IW(LIW) – INTEGER array *Workspace*
13: LIW – INTEGER *Input*

On entry: the dimension of the array IW as declared in the (sub)program from which D02JBF is called.

Constraint: LIW $\geq \mathrm{N} \times (\mathrm{K1} + 2)$.

14: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, N $<$ 1,
or X0 $\geq$ X1,
or K1 $<$ 2,
or KP $<$ K1 $-$ 1,
or IC $<$ K1.

IFAIL = 2

On entry, LW $< 2 \times$ N $\times$ (KP $+ 1) \times$ (N $\times$ K1 $+ 1) + 7 \times$ N $\times$ K1,
or LIW $<$ N $\times$ (K1 $+ 2)$ (i.e., insufficient workspace).

IFAIL = 3

Either the boundary conditions are not linearly independent, (that is, in the subroutine BC the variable J is set to the same value k for two different values of I), or the rank of the matrix of equations for the coefficients is less than the number of unknowns. Increasing KP may overcome this latter problem.

IFAIL = 4

The least-squares routine F04AMF has failed to correct the first approximate solution (see F04AMF).

7 Accuracy

The Chebyshev coefficients are determined by a stable numerical method. The accuracy of the approximate solution may be checked by varying the degree of the polynomials and the number of collocation points (see Section 8).

8 Further Comments

The time taken by the routine depends on the size and complexity of the differential system, the degree of the polynomial solution and the number of matching points.

The collocation points in the range (x_0, x_1) are chosen to be the extrema of the appropriate shifted Chebyshev polynomial. If KP = K1 $-$ 1, then the least-squares solution reduces to the solution of a system of linear equations and true collocation results.

The accuracy of the solution may be checked by repeating the calculation with different values of K1 and with KP fixed but KP $\gg$ K1 $-$ 1. If the Chebyshev coefficients decrease rapidly for each component (and consistently for various K1 and KP), the size of the last two or three gives an indication of the error. If the Chebyshev coefficients do not decay rapidly, it is likely that the solution cannot be well-represented by Chebyshev-series. Note that the Chebyshev coefficients are calculated for the range $(-1, 1)$.

Linear systems of high-order equations in their original form, singular problems, and, indirectly, nonlinear problems can be solved using D02TGF.

9 Example

To solve the equation

$$y'' + y = 1$$

with boundary conditions

$$y(-1) = y(1) = 0.$$

The equation is written as the first-order system

$$\begin{pmatrix} y_1' \\ y_2' \end{pmatrix} = \begin{pmatrix} 0 & 1 \\ -1 & 0 \end{pmatrix} \begin{pmatrix} y_1 \\ y_2 \end{pmatrix} + \begin{pmatrix} 0 \\ 1 \end{pmatrix}$$

for solution by D02JBF and the boundary conditions are written

$$y_1(-1) = y_1(1) = 0.$$

We use K1=4, 6, 8 and KP = 10 and 15, so that the different Chebyshev-series may be compared. The solution for K1 = 8 and KP = 15 is evaluated by E02AKF at nine equally spaced points over interval $(-1, 1)$.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     D02JBF Example Program Text
*     Mark 14 Revised.  NAG Copyright 1989.
*     .. Parameters ..
      INTEGER          N, K1MAX, KPMAX, IC, LW, LIW
      PARAMETER        (N=2,K1MAX=8,KPMAX=15,IC=K1MAX,LW=2*N*(KPMAX+1)
     +                 *(N*K1MAX+1)+7*N*K1MAX,LIW=N*(K1MAX+2))
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      real             X, X0, X1
      INTEGER          I, IA1, IFAIL, J, K1, KP, M
*     .. Local Arrays ..
      real             C(IC,N), W(LW), Y(N)
      INTEGER          IW(LIW)
*     .. External Functions ..
      real             CF
      EXTERNAL         CF
*     .. External Subroutines ..
      EXTERNAL         BC, D02JBF, E02AKF
*     .. Intrinsic Functions ..
      INTRINSIC        real
*     .. Executable Statements ..
      WRITE (NOUT,*) 'D02JBF Example Program Results'
      X0 = -1.0e0
      X1 = 1.0e0
      WRITE (NOUT,*)
      WRITE (NOUT,*) ' KP  K1   Chebyshev coefficients'
      DO 60 KP = 10, KPMAX, 5
         DO 40 K1 = 4, K1MAX, 2
            IFAIL = 1
*
            CALL D02JBF(N,CF,BC,X0,X1,K1,KP,C,IC,W,LW,IW,LIW,IFAIL)
*
            IF (IFAIL.NE.0) THEN
               WRITE (NOUT,99999) KP, K1, '  D02JBF fails with IFAIL =',
     +           IFAIL
               STOP
            ELSE
               WRITE (NOUT,99998) KP, K1, (C(I,1),I=1,K1)
               DO 20 J = 2, N
                  WRITE (NOUT,99997) (C(I,J),I=1,K1)
   20          CONTINUE
               WRITE (NOUT,*)
            END IF
   40    CONTINUE
   60 CONTINUE
      K1 = 8
      M = 9
      IA1 = 1
      WRITE (NOUT,99996) 'Last computed solution evaluated at', M,
     +  '  equally spaced points'
      WRITE (NOUT,*)
      WRITE (NOUT,99995) '      X ', (J,J=1,N)
      DO 100 I = 1, M
         X = (X0*real(M-I)+X1*real(I-1))/real(M-1)
         DO 80 J = 1, N
            IFAIL = 0
*
            CALL E02AKF(K1,X0,X1,C(1,J),IA1,IC,X,Y(J),IFAIL)
*
   80    CONTINUE
         WRITE (NOUT,99994) X, (Y(J),J=1,N)
  100 CONTINUE
      STOP
*
```

```
99999 FORMAT (1X,2(I3,1X),A,I4)
99998 FORMAT (1X,2(I3,1X),8F8.4)
99997 FORMAT (9X,8F8.4)
99996 FORMAT (1X,A,I3,A)
99995 FORMAT (1X,A,2('      Y(',I1,')'))
99994 FORMAT (1X,3F10.4)
      END
*
      real  FUNCTION CF(I,J,X)
*     .. Parameters ..
      INTEGER          N
      PARAMETER        (N=2)
*     .. Scalar Arguments ..
      real             X
      INTEGER          I, J
*     .. Local Arrays ..
      real             A(N,N), R(N)
*     .. Data statements ..
      DATA             A(1,1), A(2,1), A(1,2), A(2,2)/0.0e0, -1.0e0,
     +                 1.0e0, 0.0e0/
      DATA             R(1), R(2)/0.0e0, 1.0e0/
*     .. Executable Statements ..
      IF (J.GT.0) CF = A(I,J)
      IF (J.EQ.0) CF = R(I)
      RETURN
      END
*
      SUBROUTINE BC(I,J,RHS)
*     .. Scalar Arguments ..
      real           RHS
      INTEGER        I, J
*     .. Executable Statements ..
      RHS = 0.0e0
      IF (I.GT.1) THEN
         J = -1
      ELSE
         J = 1
      END IF
      RETURN
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 D02JBF Example Program Results

  KP  K1   Chebyshev coefficients
  10   4  -0.7798 -0.0000  0.3899  0.0000
           0.0000  1.5751 -0.0000 -0.0629

  10   6  -0.8326 -0.0000  0.4253  0.0000 -0.0090 -0.0000
           0.0000  1.6290  0.0000 -0.0724 -0.0000  0.0009

  10   8  -0.8325 -0.0000  0.4253  0.0000 -0.0092 -0.0000  0.0001  0.0000
           0.0000  1.6289  0.0000 -0.0724 -0.0000  0.0009  0.0000 -0.0000

  15   4  -0.7829  0.0000  0.3914 -0.0000
          -0.0000  1.5778  0.0000 -0.0631

  15   6  -0.8326  0.0000  0.4253 -0.0000 -0.0090 -0.0000
          -0.0000  1.6290 -0.0000 -0.0724 -0.0000  0.0009

  15   8  -0.8325  0.0000  0.4253 -0.0000 -0.0092 -0.0000  0.0001  0.0000
          -0.0000  1.6289 -0.0000 -0.0724 -0.0000  0.0009  0.0000 -0.0000

 Last computed solution evaluated at  9  equally spaced points
```

```
     X        Y(1)      Y(2)
  -1.0000    0.0000   -1.5574
  -0.7500   -0.3542   -1.2616
  -0.5000   -0.6242   -0.8873
  -0.2500   -0.7933   -0.4579
   0.0000   -0.8508   -0.0000
   0.2500   -0.7933    0.4579
   0.5000   -0.6242    0.8873
   0.7500   -0.3542    1.2616
   1.0000    0.0000    1.5574
```